/*
 * Copyright the original author or authors.
 * 
 * Licensed under the MOZILLA PUBLIC LICENSE, Version 1.1 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 * 
 *	  http://www.mozilla.org/MPL/MPL-1.1.html
 * 
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
 package com.googlecode.milib.display.button.views {
	import com.googlecode.milib.display.button.views.AbstractButtonView;
	import com.googlecode.milib.display.service.DisplayObjectFiltersService;
	import com.gskinner.geom.ColorMatrix;

	import flash.display.DisplayObject;
	import flash.filters.ColorMatrixFilter;
	import flash.utils.Dictionary;

	/**
	 * @author Marek Brun
	 */
	public class ButtonViewContrast extends AbstractButtonView {
		
		private var display:DisplayObject;
		private static const servicedObjects : Dictionary = new Dictionary(true);
		private var cMatrixOver:ColorMatrix;
		private var cMatrixPress:ColorMatrix;
		private var filters:DisplayObjectFiltersService;

		
		public function ButtonViewContrast(access:Private, display:DisplayObject) {
			this.display=display;
			cMatrixOver=new ColorMatrix((new ColorMatrixFilter()).matrix);
			cMatrixOver.adjustContrast(15);
			cMatrixPress=new ColorMatrix((new ColorMatrixFilter()).matrix);
			cMatrixPress.adjustContrast(30);
			filters=DisplayObjectFiltersService.forInstance(display);
		}
		
		override protected function doCurrentState():void {
			switch(state){
				case AbstractButtonView.state_OUT:
					filters.setFiltersFromID(this, []);
				break;
				case AbstractButtonView.state_OVER:
					filters.setFiltersFromID(this, [new ColorMatrixFilter(cMatrixOver)]);
				break;
				case AbstractButtonView.state_PRESSED:
					filters.setFiltersFromID(this, [new ColorMatrixFilter(cMatrixPress)]);
				break;
				case AbstractButtonView.state_DISABLED:
					
				break;
				default:
					throw new ArgumentError('Unexpected state object');
			}
		}
		
		public static function forInstance(display:DisplayObject):ButtonViewContrast {
			if(servicedObjects[display]){
				return servicedObjects[display];
			}else{
				servicedObjects[display]=new ButtonViewContrast(null, display);
				return servicedObjects[display];
			}
		}
		
	}
}
internal class Private {}